TEMPLATE = app
LANGUAGE = C++
CONFIG += qt resources
INCLUDEPATH += ../../libniwbillmanager/includes
QMAKE_EXTRA_UNIX_TARGETS += launcher

include(../../../paths.pri)
include(../../../settings.pri)

isEqual(DEBUG, 1) {
    CONFIG += debug
}
else {
    CONFIG += release
}

MOC_DIR = ../../../build/moc
UI_DIR = ../../../build/ui
OBJECTS_DIR = ../../../build
RCC_DIR = ../../../build/rcc

includes.files += plugins/qt4uiplugininterface.h

target.path = $${BINDIR}
data.path = $${SHAREDIR}
includes.path = $${INCLUDEDIR}

launcher.files = launcher/niw-bill-manager.sh
launcher.path = $${BINDIR}

isEqual(INSTALL_QT4_LAUNCHER, 1) {
    INSTALLS += launcher
}

DESTDIR = ../../../
TARGET = niw-bill-manager
INSTALLS += target data includes


FORMS += forms/mainwindow.ui \
         forms/aboutdialog.ui \
         forms/billitemdialog.ui \
         forms/pluginbrowserdialog.ui \
         forms/remotelocationdialog.ui \
         forms/settingsdialog.ui

HEADERS = mainwindow.h \
          ../../libniwbillmanager/includes/niwbillmanager.h \
          ../../constants.h \
          plugins/qt4uiplugininterface.h \
          aboutdialog.h \
          billitemdialog.h \
          plugin.h \
          pluginbrowserdialog.h \
          pluginmanager.h \
          remotelocationdialog.h \
          settingsdialog.h

SOURCES = main.cpp \
          mainwindow.cpp \
          ../../constants.cpp \
          aboutdialog.cpp \
          billitemdialog.cpp \
          plugin.cpp \
          pluginbrowserdialog.cpp \
          pluginmanager.cpp \
          remotelocationdialog.cpp \
          settingsdialog.cpp

RESOURCES += resources/resources.qrc \
             resources/icons.qrc

LIBS = -L../../../lib -lniwbillmanager
